ABLATION, AEROBRAKING AND AIRBURSTING OF A HYPERSONIC PROJECTILE IN EARTH'S ATMOSPHERE

PAUL J. THOMAS
MARC GOULET
ANDREW T. PHILLIPS
ALEX SMITH

1. Overview

This module is intended as a stand-alone component of a second, project-based course in computational science. The students should have two semesters of calculus and interest in physics, astronomy or geology. It assumes some proficiency with the symbolic and programming capabilities of Maple, as might be taught in a first course in computational science. The module is implemented in its entirety using Maple.

The learning goals are as follows:

- To become exposed to coupled differential equations, by studying the equations of motion, ablation and deformation in the context of a hypersonic projectile moving through Earth's atmosphere.
- To analytically solve a restricted solution for this problem.
- To develop the ability to use an explicit time differencing scheme to solve the complete set of coupled differential equations.
- To construct and interpret graphs so as to visualize aspects of the dynamic evolution of a hypersonic projectile in Earth's atmosphere, and to compare model results to data obtained from the analysis of such objects.

2. Introduction to the Problem

All major bodies in the solar system have been shaped by a continuous rain of impacting objects from space. These impactors are composed of material left over from the formation of the planets, and the rate of impact was extremely high during the early history of the solar system, a period of time 4.5–3.8 Gya (billion years ago) known as the **Heavy Bombardment era**. The craters formed by impacts during the Heavy Bombardment era are common on all planetary objects where erosion and geological activity has been sufficiently small for these ancient features to survive: Mercury, the highlands of the Earth's Moon, the southern hemisphere of Mars and many of the satellites of the outer solar system.

Over the last three billion years, the surface of the Earth has continued to be bombarded by objects of various sizes, mostly asteroids from the asteroid belt between Mars and Jupiter. Their orbits have been altered by the gravitational influence of the giant planets, principally Jupiter. Typical collision velocities for these objects with the Earth are 15 km/s. In addition to asteroids, comets from the outer solar system (the Kuiper Belt and Oort Cloud) can also impact with planets. Because their trajectories originate at much greater distances, their impact velocities are accordingly greater, ranging between 25 and 50 km/s.

When an asteroid or a comet encounters a planetary atmosphere, a complex set of physical processes occur. It is important to recognize that the velocities described above are greatly in excess of the speed of sound ($\sim$0.3 km/s) and so we call these objects **hypersonic projectiles**. In addition to these natural objects, artificial hypersonic projectiles include returning manned spacecraft such as the U.S. Space Shuttle and the Russian Soyuz, ICBM warheads and abandoned space stations (e.g. Mir and Skylab).

The principal physical processes experienced by a hypersonic projectile are **atmospheric drag** (the reduction of velocity caused by atmospheric drag forces), **ablation** (the reduction in mass caused by evaporation of the projectile due to atmospheric heating) and **deformation** (the change in shape of the projectile caused by differential negative acceleration forces). These forces vary during the passage of the projectile through the atmosphere, both because the projectile's speed varies but also because the atmospheric density increases sharply as the projectile travels closer to the surface. In extreme cases, the projectile may deform sufficiently greatly to produce an **airburst explosion**, where all of its kinetic energy is lost in a very short period of time. Such an event occurred in Tunguska, Siberia, in 1908. The airburst here caused an atmospheric shockwave that devastated a 2,200 km^2 area of pine forest [*Zotkin and Tsikulin*, 1966].

3. Statement of the Problem

In this module you will learn how to numerically integrate differential equations describing the hypersonic passage of a projectile through the Earth's atmosphere.

The first set of differential equations describe the forces acting on the projectile: atmospheric drag, gravity and atmospheric lift. Atmospheric drag, as the name implies, acts to slows the projectile, while gravity will accelerate the projectile towards the center of the Earth. This latter effect will tend not only to increase the speed of the projectile but will also change its direction of motion towards a radius line from the center of the Earth. Finally, atmospheric lift will tend to change a projectile's direction of motion towards a more horizontal trajectory. This effect will be small for most of the objects that we consider here, but it can be important: the "Great Daylight 1972 Wyoming Fireball" was a meteor that generated sufficient lift during its 100 minute passage through the upper atmosphere that it "skipped" back into space! [*Rawcliffe et al.*, 1974; *Ceplecha*, 1994].

The second differential equation describes ablative heating. This effect is the extreme heating that a hypersonic projectile experiences as it passes through the atmosphere. The atmosphere in the vicinity of the projectile can reach temperatures hotter than the Sun's surface (6,000 °C), which will cause the surface of the projectile to evaporate (ablate). Ablation in turn affects atmospheric drag, as the amount of drag is dependent on the cross-sectional area of the projectile, which is reduced by ablation.

The third differential equation describes deformation of the projectile. This is caused by the extreme pressures due to the projectile's rapid motion through the atmosphere. The effect of deformation is to cause the projectile to assume a "pancake" shape, which can drastically increase atmospheric drag. In some cases, this process can cause an airburst explosion.

The focus of this module is to examine the projectile behavior governed by these equations for various projectile types (carbonaceous, stony and iron asteroids and comets) to address the following questions:

(1) How does the velocity of the projectile vary with altitude and time as it passes through the atmosphere?
(2) How much of the projectile (if any) survives to impact the surface?
(3) Does the projectile airburst explosively?

4. Background Information

4.1. **Forces due to Atmospheric Drag and Gravity.**



Figure 1. Geometry of the projectile model.

Initially, we will consider the trajectory equations, given by

$$m\frac{dv}{dt} = -\frac{1}{2}C_D\rho_a Av^2 + mg\sin\theta \tag{1}$$

and

$$\frac{d\theta}{dt} = \frac{g\cos\theta}{v} - \frac{C_L\rho_a Av}{2m} - \frac{v\cos\theta}{R_P + z}. \tag{2}$$

Note that the right hand side in equation (1) consists of the sum of two terms. The first term is due to atmospheric drag and the second term contains a component due to gravity. The second equation contains three terms. The first term contains a gravity component (the orthogonal one to that in equation (1). The second term is due to atmospheric lift and the third term accounts for the sphericity of the Earth's gravitational acceleration. In these equations, $v = v(t)$ represents the velocity of the projectile as a function of time t. The quantities $m = m(t)$, and $A = A(t)$ represent the mass and cross-sectional area of the projectile. θ is the angle of the trajectory, measured from the horizontal (so a vertical trajectory would be 90°). C_D and C_L are the coefficients of drag and lift, respectively, and depend on the shape of the projectile. R_P is the radius of the Earth, z the altitude above the Earth's surface and $g = g(z)$ and $\rho_a = \rho_a(z)$ the gravitational acceleration and atmospheric density at that height.

For this problem, we have to take account of variations of both gravity and atmospheric density with height. Gravity varies with height as

$$g(z) = g_0 \left(\frac{R_P - z}{R_P} \right)^2 \tag{3}$$

where $g_0 = 9.81$ m s^{-2}, the acceleration due to gravity at the surface of the Earth, and $R_P = 6.371 \times 10^6$ m, the radius of the Earth.

Atmospheric density varies with height as

$$\rho_a(z) = \rho_0 e^{-z/H} \tag{4}$$

where $\rho_0 = 1.22$ kg m^{-3}, the surface atmospheric density and $H = 8100$ m, the scale height for the Earth's atmosphere. (This term, which measures the height over which the Earth's atmosphere decreases in density by a factor of $1/e$, itself varies with height, but we will ignore that for this exercise). A superior model for atmospheric density would use the tabulated data from the *US Standard Atmosphere* (1976), which could be embedded in an interpolation routine to provide density values at any height.

4.2. **Ablation.** We now consider the effect of ablative heating on the projectile. Since the projectile is hypersonic, its motion is so fast that the air in front of it does not have sufficient time to move out of the way. Instead, the air is compressed into a dense layer whose density and pressure is discontinuous from the surrounding atmosphere. This discontinuity is called a **shock front**. Temperatures in the shock front can easily reach 6,000 K, similar to the temperature of the surface of the Sun. Naturally, the air in the shock front is transformed into an ionized plasma. Radiation from the shock front heats the projectile and caused evaporation, or ablation of the projectile surface. Early research during the space age [*Allen and Eggers*, 1958] indicated that heating of spacecraft during entry into a planetary atmosphere would be survivable if the shock front was kept as far from the spacecraft as possible. This could be achieved by making the spacecraft shape into a blunt, non-aerodynamic barrier, exemplified by the shapes of the Apollo and Soyuz heat shields or the underside of the space shuttle.

Analysis of the ablation of meteors reveals that the mass loss of a projectile due to heating by the shock front is given by

$$Q\frac{dm}{dt} = -\frac{1}{2}C_H\rho_a A v^3 \tag{5}$$

where Q is the heat of ablation and C_H is the heat transfer coefficient [*Bronshten*, 1983]. Q is a function of material type and the specific process of ablation. Values of Q for various types of asteroids and comets are given in Table 1 (from *Chyba et al.*, 1990).

Table 1: Parameters for various different atmospheric projectiles

Object type	Density (kg m−3)	Velocity (m s^{-1})	Heat of ablation (MJ kg^{-1})	Yield strength (MPa)
Iron	7900	15,000	8.0	100
Stone	3500	15,000	8.0	10
Carbonaceous	2200	15,000	5.0	1
SP Comet	1000	25,000	2.5	0.1

The ablation equation above describes mass loss accurately for higher altitudes in the atmosphere, where most visible meteors ablate, leaving behind "meteor trails" [*Bronshten*, 1983]. However, at lower altitudes ($z < 30$ km), an increasing amount of projectile kinetic energy is absorbed by creating more ionized gas at the shock front. Typically the shock front temperature has a maximum value of 25,000 K and so the ablation rate under these conditions is limited to a maximum rate of

$$Q\frac{dm}{dt} = -A\sigma T_{max}^4 \tag{6}$$

where $\sigma = 5.670 \times 10^{-8}$ J m^{-2} s^{-1} K^{-4} is Stefan's constant and $T_{max} = 25,000$ K [*Biberman et al.*, 1980]. A general ablation mass loss equation should therefore be written as

$$Q\frac{dm}{dt} = -A\min\left(\frac{1}{2}C_H\rho_a v^3, \sigma T_{max}^4\right) \tag{7}$$

We can determine that $C_H = 0.1$ from photographic observation of meteors [*Bronshten*, 1983].

4.3. **Deformation and Fragmentation.**



Figure 2. Geometry of the deformation model.

We now consider the effect of atmospheric drag on the shape of the projectile. As discussed in the previous section, the hypersonic motion of the projectile causes a shock front of air to build up in front of the projectile. We can estimate how large is pressure is by considering the drag force from equation (1). Examining the first term, we see that the magnitude of the net atmospheric drag force on the projectile is given by

$$F_{net} = \frac{1}{2}C_D\rho_a A v^2. \tag{8}$$

Since force equals pressure times area, the average pressure P_s on the front surface of the projectile is

$$P_s = \frac{1}{2} C_D \rho_a v^2. \tag{9}$$

Similarly, the air pushed out of the way by the passage of the projectile does not have time to flow back into the volume behind it. As a result, the projectile has a high pressure exerted on its leading face and a vacuum at its trailing face. This difference in pressure causes the deformation and fragmentation of many objects entering the Earth's atmosphere.

In our model (following the work of *Chyba et al.*, 1993), we will assume that fragmentation occurs when the leading face pressure P_s exceeds the material strength of the projectile. Material strengths for projectiles of various types can be determined by laboratory experiments on meteorites and observations of meteors and are listed in Table 1. As a comparison, note that atmospheric pressure at sea level is approximately 0.1 MPa.

Fragmentation is a complex process, and can only be analyzed in detail with a numerical model that allows the turbulent behavior of the atmosphere to be represented. A good recent work that shows this type of modeling is *Korycansky and Zahnle* (2003). Such modeling requires substantial amount of time on supercomputers or Beowulf clusters, so we will consider a simpler approach here.

When a projectile fragments due to atmospheric pressure, it "pancakes", so that a roughly spherical solid object becomes a flat layer of rubble. This layer has a greatly increased surface area A, and so experiences far higher drag forces than the original projectile (consider the effect of increasing A in equation (1)). As drag forces increase, the deformation of the projectile increases as well, leading to a exponentially reinforcing process that causes the projectile to be stopped in a very short distance, releasing its kinetic energy in an airburst.

A simple model of this catastrophic deformation process has been constructed by *Chyba et al.* (1993). This model assumes the projectile to be initially a cylinder, with height h equal to diameter $2r$. The symmetry axis of the cylinder is oriented in the direction of motion of the projectile. The reason for the shape is not that it is likely to be found in nature (it certainly isn't!) but that it makes deformation easier to calculate. This model turns out to give results that are consistent with observations of meteorite impacts and airbursts (*Chyba et al.*, 1993).

This model assumes that, once the critical pressure has been reached, the cylinder fractures to become a squatter version of itself. If the pressure at the leading face is P_s and the pressure at the trailing face is 0, the average pressure inside the projectile is $P_s/2 = C_D\rho_a v^2/4$. We can then calculate the radius change of the cylinder r by calculating $F = ma$ on the side walls of the cylinder as follows:

$$\text{(Pressure)} \times \text{(Side area)} = \left(\frac{1}{4}C_D\rho_a v^2\right)\left(2\pi r h\right) = m\frac{d^2r}{dt^2}. \tag{10}$$

If we assume that the projectile's density ρ_m remains constant, we can simplify the above equation to

$$r\frac{d^2r}{dt^2} = \frac{C_D\rho_a v^2}{2\rho_m}. \tag{11}$$

5. Developing and Implementing the Model

5.1. **Analytic Solution of Simple Atmospheric Drag Model.** Before we attempt to solve the complete set of differential equations in a numerical way, we will use Maple as a symbolic manipulating tool to solve the atmospheric drag equation for a simple, restricted case. Work the project found in 7.1.

5.2. **Finite Differencing of the Equations of Motion.** We have now assembled all of the equations that govern the behavior of a projectile passing through the atmosphere at hypersonic speeds. The projectile equations (1) and (2) can be solved using analytic techniques as long as mass m is held constant. However, we are interested in cases where ablation and airbursting can occur. Therefore the equations need to be solved numerically, through finite differences [*Press et al.*, 1992].

As you know, a good approximation to the derivative is provided by the difference quotient. So, for a function f that depends on time t we can write

$$\frac{df}{dt} \approx \frac{f(t+\Delta t) - f(t)}{\Delta t}.$$

If Δt is sufficiently small, this is a good approximation. For numerical purposes, the difference quotient replaces the derivative. Sometimes we use a so-called central difference to approximate the derivative:

$$\frac{df}{dt} \approx \frac{f(t+\Delta t) - f(t-\Delta t)}{2\Delta t}.$$

Think about averaging the left and right derivatives.

Similarly, the second derivative can be approximated using finite differences by

$$\frac{d^2 f}{dt^2} \approx \frac{f'(t+\Delta t) - f'(t)}{\Delta t} \approx \frac{f(t+\Delta t) - 2f(t) + f(t-\Delta t)}{\Delta t^2}.$$

For our problem we are interested in the unknowns $v(t)$, $\theta(t)$ and $m(t)$. We need to discretize t. Using the form for the derivative above, we can write equation (1) in the form

$$v_{new} = v_{old} - \frac{C_D \rho_a A v_{old}^2 \Delta t}{2m} + g \sin\theta_{old} \Delta t, \tag{12}$$

where we can think about v_{new} as being $v(t+\Delta t)$ and v_{old} as being $v(t)$. Note that the updated velocity v_{new} is smaller than the previous velocity v_{old} due to a negative term containing atmospheric drag and larger due to a positive term containing gravitational acceleration. This should make physical sense.

Look closely at the equation above. Using this approach, it should be straightforward to write equations for similar updated values for θ_{new} and m_{new}. Derive these equations.

We can add one more equation to those that you have just derived, that keeps track of decreasing altitude z:

$$z_{new} = z_{old} - v_{old} \sin\theta_{old} \Delta t. \tag{13}$$

Now we are ready to build our first numerical model. Work the project found in 7.2.

5.3. **Adding Ablation.** Adapt your Maple worksheet to include ablation. For the sample case discussed above, how do the results change? Explain, in physical terms, the change in behavior that you see. Work the project found in 7.3.

5.4. **Adding Deformation.** Now adapt your Maple worksheet to include deformation and fragmentation. You will want to write a section of the code that solves the differential equation in r only when the stagnation pressure P_s exceeds a critical value determined by the projectile's strength. Work the project found in 7.4.

6. Conceptual Questions and Supplemental Projects

(1) Can you see a relationship between the altitudes at which the projectile deforms and where the maximum kinetic energy is deposited in the atmosphere? Explain this relationship.
(2) Examine all of the equations describing the motion, ablation and deformation of the projectile and describe the physical processes that occur during an airburst.
(3) The iron candidate Tunguska object impacted the Earth's surface without airbursting. This object is a good candidate for the Barringer Meteor Crater in Arizona. Using library and internet research, compare the two features and discuss whether an airburst is likely to be more devastating than an impact of the same energy.
(4) You can use the model developed in this module to examine the fates of comets and asteroid projectiles on other planets. To do this, you need to change the parameters for ρ_a, H, g_0 and R_P. The table below lists values of these parameters for the atmospheres of Venus, Mars and Jupiter (*NASA Planetary Atmospheres Node*, 2003). Note that the density ρ_0 for Jupiter is not given for a solid surface (Jupiter doesn't have one), but for a "reference surface" where the atmospheric pressure is equal to 1 bar, the pressure at the surface of the Earth.

Table 2: Model Parameters for Other Planets

Planet	ρ_0 (kg m^{-3})	H (m)	g_0 (m s^{-2})	R_P (m)
Venus	6.662×10^{-2}	15,519	8.93	6.052×10^6
Mars	1.731×10^{-5}	10,807	3.73	3.397×10^6
Jupiter	1.618×10^{-4}	25,476	23.2	7.1492×10^7

Using this extended model, answer the following questions.

(5) Are airburst explosions more common on Venus than on the Earth? For example, consider a set of projectiles that do impact the Earth (say, iron and stone asteroids twice as large as the Tunguska candidates). Do they airburst in Venus's atmosphere?

(6) Use your model to calculate the airburst altitudes of the fragments of Comet Shoemaker–Levy 9 in Jupiter's atmosphere. To model this case, the initial projectile should be $61,000$ m/s and the projectiles should be comets with radii of 500 m.

7. Problems and Projects

7.1. Project 1: Symbolic Solution of Simplified Equations of Motion.

We can symbolically solve the equations of motion for the projectile for simplified cases where ablation and deformation is not taken into account. In this project we'll discuss two such cases.

(1) For the first case, we'll assume that the projectile is moving vertically towards the Earth and that gravity and atmospheric density are constant (specifically, that they have their surface values: $g = 9.8$ m s^{-2} and $\rho_a = 1.22$ kg m^{-3}. For this case, equation (1) simplifies to

$$m\frac{dv}{dt} = -\frac{1}{2}C_D\rho_a Av^2 + mg \tag{14}$$

Use Maple in symbolic mode to find an analytic solution to this equation and obtain a graph of speed v vs. altitude z. A good approach would be as follows. We can convert equation (14) to a first order equation by using the identity

$$m\frac{dv}{dt} = mv\frac{dv}{dz} = -\frac{1}{2}C_D\rho_a Av^2 + mg \tag{15}$$

This equation is now in a form that Maple's `dsolve` utility can solve. Two solutions will be found (select the positive one), and a constant of integration will need to be evaluated. Use the initial condition $v(z = z_0) = v_0$. Numerical values that we can use to obtain a graph are $C_D = 1.7$, $A = \pi$, $m = 2\pi$, $x_0 = 100000$ m, $v_0 = 15000$ m s^{-1}.

(2) For the second case, we'll assume that atmospheric density varies with height as given by equation (4). As before, use Maple in symbolic mode to find an analytic solution to equation (15) and obtain a graph of speed v vs. altitude z.

(3) Discuss the difference between the two graphs.

7.2. **Project 2: Numerical Solution of the Equations of Motion.** In this first numerical model, we will ignore ablation and deformation, but simply consider the behavior of the projectile due to drag and gravitational acceleration. In order to use this finite differencing approach, we need initial values for the variables.

The drag coefficient $C_D = 1.7$ for the cylindrical shapes we are considering (it does not change much for other shapes). Observations of meteor behavior indicates that C_L=0.001 [*Passey and Melosh*, 1980]. We will start our simulation at an altitude of 100 km. The value of g at this altitude (from equation (4)) is 9.50 m s^{-2}.

The initial value for mass can be taken from Table 1, which lists properties for various different types of asteroids and comets. Let's assume a stony asteroid (v = 15,000 m s^{-1} and ρ_m = 3500 kg m^{-3}), an initial velocity angle 45° with an initial radius of 10 m. With this information, we can calculate the initial mass. To be consistent with the cylindrical model we will eventually use for deformation, we should obtain it from

$$m = \left(\pi r^2\right)\left(2r\right)\rho_m. \tag{16}$$

We can obtain the cross sectional area A similarly

$$A = \pi r^2. \tag{17}$$

Both of the equations above will yield the changing values of m and A from the current values of r_{new}.

A projectile moving with a speed of 15,000 m s^{-1} will cover a distance of 100 km in about 6.6 s. Given that the projectile will slow due to drag forces, the total simulation time is likely to be approximately 10 s. We therefore choose a time interval of 0.01 s. This choice likely to yield about 1000 timesteps, a number that will not require a very long computational time on a standard computer, but is likely to produce accurate and stable numerical results for this case. (A more thorough approach would use an implicit numerical scheme, such as the Fourth-Order Runge-Kutta method).

To summarize, the algorithm for the Maple program you need to write is as follows

(1) Assign initial values to all variables: m, v, θ, A, g, z, ρ_a, elapsed time $t = 0$.
(2) Assign values to all constants: C_D, C_L, R_P.
(3) Start loop. The loop variable is time t, which you increment by Δt for each step. The loop ends when the projectile reaches the surface of the Earth ($z = 0$). You might want to consider using the *while* loop construct in Maple.
(4) Inside the loop, obtain new values for m, v, θ, A, g, z, ρ_a and t. (Note that, for this first model, m and A will be constant as we ignore both ablation and deformation).
(5) As you calculate values for m, v θ and z , enter the results in arrays so that you can plot them.

Plot z vs. t, v vs. z, θ vs. z and m vs. z. Explain your results, based on the physical concepts we've discussed in this module.

7.3. **Project 3: Numerical Solution of the Equations of Motion with Ablation.** The Maple program you have used before can be adapted to incorporate ablation. Use equation (5) in finite difference form to modify mass m as the projectile passes through the atmosphere. Note that cross sectional area A changes as mass changes.

For this model and the ones that follow, we will track a parameter that is of special interest: **the kinetic energy change with height**. At each step, calculate kinetic energy (by calculating the current value of $KE = 1/2mv^2$ and calculate the following variable:

$$\frac{\Delta KE}{\Delta z} = \frac{KE_{new} - KE_{old}}{z_{new} - z_{old}} \tag{18}$$

Plot v vs. z, m vs. z and $\Delta KE/\Delta z$ vs z. Explain your results, based on the physical concepts we've discussed in this module.

7.4. **Project 4: Numerical Solution of the Equations of Motion with Ablation and Deformation.** You can further adapt your Maple program to incorporate deformation. It is this mechanism that causes airbursts, where most of the kinetic energy is deposited in a narrow range of altitudes. Obviously the graph of kinetic energy change with height is of special interest in this problem.

The deformation mechanism is implemented as follows. First, we need to track the stagnation pressure in the finite difference loop by continuously recalculating equation (9). When the stagnation pressure exceeds material strength, you need to begin to calculate the spreading of the projectile for each timestep. This is done by a finite difference version of equation (11), which is

$$r(t+\Delta t) = 2r(t) - r(t-\Delta t) + \frac{C_D \rho_a(z(t-\Delta t))v(t-\Delta t)^2 \Delta t^2}{2\rho_m r(t)} \tag{19}$$

This equation must be included in the finite differencing only when the stagnation pressure has exceeded material strength. One way to do this is to set a variable (for example, one called *fragment*) to zero initially. As long as *fragment* remains equal to zero, changes in radius are calculated from ablation only. As soon as the stagnation pressure exceeds material strength, set the value of *fragment* to 1. If the deformation code can only execute once *fragment* equals 1, it will now start to calculate the changing radius of the projectile.

To summarize, an algorithm for this part of the worksheet might look like this:

(1) At beginning of worksheet, set *fragment* to 0.
(2) If stagnation pressure is less than material strength and *fragment* equals 0, calculate change in projectile radius from ablation calculation only. If *fragment* equals 1, then calculate change in radius from equation (19).
(3) If stagnation pressure exceeds material strength, then set *fragment* equal to 1.
(4) Continue the finite difference loop.

Implement the deformation procedure in your Maple model. Use the data for a stone asteroid from Table 1. Assume that the asteroid is 10 m in radius initially. Plot v vs. z, m vs. z, $\Delta KE/\Delta z$ vs z and z vs. r. Explain the shapes of the graphs you obtain.

Can we define a specific airburst altitude for this projectile? If so, what value would it have?

7.5. **Project 5: Tunguska.** The explosion over the Tunguska river valley in central Siberia on June 30, 1908 is estimated to have released approximately 6×10^{16} J of energy, roughly the explosive yield of a 15 megaton hydrogen bomb. This energy was not released at the surface, as no crater was found. Instead the explosion toppled tree trunks over a 2200 km^{-2} area in a pattern radiating from a central point. Modeling of the treefall pattern indicates that the explosion occurred at an altitude of 10 km [*Zotkin and Tsikulin*, 1966].

The table below lists a set of candidate objects that contain the correct amount of initial kinetic energy to account for the Tunguska explosion. Which object releases the majority of its kinetic energy at the appropriate altitude of 10 km?

Table 3: Candidate Tunguska projectiles

Object	Density ($\mathrm{kg\ m^{-3}}$)	Radius (m)	Velocity (m/s)	θ
Iron	7900	22	15	45°
Stone	3500	29	15	45°
Carbonaceous	2200	34	15	45°
Comet	1000	32	25	45°

8. Solutions

8.1. **Problem 7.1: Symbolic Solution of Simplified Equations of Motion (1).**

This is for constant atmospheric density. Define the de in the following first order form:

$$deq1 := m\,\mathrm{v}(z)\,(\tfrac{d}{dz}\,\mathrm{v}(z)) - k\,\mathrm{v}(z)^2 + m\,g = 0$$

Then, solve the de. Two solutions will be found. Note the constant of integration, _C1. We need to solve for that using initial conditions.

```
> soln1:=dsolve(deq1,v(z));
```

$$soln1 := \mathrm{v}(z) = \frac{\sqrt{k\,(m\,g + e^{(\frac{2\,k\,z}{m})}\,\_C1\,k)}}{k},\ \mathrm{v}(z) = -\frac{\sqrt{k\,(m\,g + e^{(\frac{2\,k\,z}{m})}\,\_C1\,k)}}{k}$$

Select the positive one.

```
> soln1:=soln1[1];
```

$$soln1 := \mathrm{v}(z) = \frac{\sqrt{k\,(m\,g + e^{(\frac{2\,k\,z}{m})}\,\_C1\,k)}}{k}$$

Now solve for _C1 using initial conditions. Our initial condition is that v(z=z0)=v0.

```
> eq1:=eval(rhs(soln1),z=z0)=v0;
```

$$eq1 := \frac{\sqrt{k\,(m\,g + e^{(\frac{2\,k\,z0}{m})}\,\_C1\,k)}}{k} = v0$$

Now find evaluate _C1 for the initial conditions.

```
> solve({eq1},{_C1});
```

$$\left\{\_C1 = -\frac{m\,g - v0^2\,k}{e^{(\frac{2\,k\,z0}{m})}\,k}\right\}$$

Now substitute this value of _C1 into the solution to the de.

```
> soln1:=eval(soln1,%);
```

$$soln1 := \mathrm{v}(z) = \frac{\sqrt{k\left(m\,g - \frac{e^{(\frac{2\,k\,z}{m})}\,(m\,g - v0^2\,k)}{e^{(\frac{2\,k\,z0}{m})}}\right)}}{k}$$

```
> soln1:=rhs(soln1);
```

$$soln1 := \frac{\sqrt{k\left(m\,g - \frac{e^{\left(\frac{2\,k\,z}{m}\right)}\,(m\,g - v0^2\,k)}{e^{\left(\frac{2\,k\,z0}{m}\right)}}\right)}}{k}$$

Now plot the results. Specify numerical values of constants.

```
> Cd:=1.7; A:=Pi; m:=2*Pi; rho0:=1.22;
  g:=9.81:  H:=8100.0;
  z0:=100000.0; v0:=15000.0; k:= 0.5*Cd*A*rho0*exp(-z0/(2*H));
> plot(soln1,z=0..z0,title="Speed vs.
  Altitude",thickness=3,labels=["z(m)","v(m/s)"]);
```



8.2. **Problem 7.1: Symbolic Solution of Simplified Equations of Motion (2).**

Now let's make it a little more interesting by letting k be dependent on z, through atmospheric density dependence. Here the drag term is -0.5 Cd A rho0 exp(-z/H). Gravity is constant. The solution technique is otherwise the same as before.

```
> deq2:=m*v(z)*Diff(v(z),z)-0.5*Cd*A*rho0*exp(-z/H)*v(z)^2+m*g=0;
```

$$deq2 := m\,\mathrm{v}(z)\,(\tfrac{d}{dz}\,\mathrm{v}(z)) - 0.5\,Cd\,A\,\rho 0\,e^{(-\frac{z}{H})}\,\mathrm{v}(z)^2 + m\,g = 0$$

```
> soln2:=dsolve(deq2,v(z));
```

$$soln2 := \mathrm{v}(z) = \frac{\sqrt{-e^{\%1}\,(2\,g\,H\,\mathrm{Ei}(1,\,-\%1) - \_C1)}}{e^{\%1}},\ \mathrm{v}(z) = -\frac{\sqrt{-e^{\%1}\,(2\,g\,H\,\mathrm{Ei}(1,\,-\%1) - \_C1)}}{e^{\%1}}$$

$$\%1 := \frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z}{H})}}{m}$$

```
> soln2:=soln2[1];
```

$$soln2 := \mathrm{v}(z) = \frac{\sqrt{-e^{\left(\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z}{H})}}{m}\right)}\left(2\,g\,H\,\mathrm{Ei}\left(1,\,-\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z}{H})}}{m}\right) - \_C1\right)}}{e^{\left(\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z}{H})}}{m}\right)}}$$

```
> eq2:=eval(rhs(soln2),z=z0)=v0;
```

$$eq2 := \frac{\sqrt{-e^{\left(\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z0}{H})}}{m}\right)}\left(2\,g\,H\,\mathrm{Ei}\left(1,\,-\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z0}{H})}}{m}\right) - \_C1\right)}}{e^{\left(\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z0}{H})}}{m}\right)}} = v0$$

```
> solve({eq2},{_C1});
```

$$\left\{\_C1 = 2\,g\,H\,\mathrm{Ei}\left(1,\,-\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z0}{H})}}{m}\right) + v0^2\,e^{\left(\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z0}{H})}}{m}\right)}\right\}$$

```
> soln2:=eval(soln2,%);
```

$$soln2 := \mathrm{v}(z) = \left(-\,e^{\left(\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z}{H})}}{m}\right)}\left(2\,g\,H\,\mathrm{Ei}\left(1,\,-\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z}{H})}}{m}\right) - 2\,g\,H\,\mathrm{Ei}\left(1,\,-\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z0}{H})}}{m}\right) - v0^2\,e^{\left(\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z0}{H})}}{m}\right)}\right)\right)^{(1/2)} \Big/ e^{\left(\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z}{H})}}{m}\right)}$$

```
> soln2:=rhs(soln2);
```

$$soln2 := \left(-\,e^{\left(\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z}{H})}}{m}\right)}\left(2\,g\,H\,\mathrm{Ei}\left(1,\,-\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z}{H})}}{m}\right) - 2\,g\,H\,\mathrm{Ei}\left(1,\,-\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z0}{H})}}{m}\right) - v0^2\,e^{\left(\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z0}{H})}}{m}\right)}\right)\right)^{(1/2)} \Big/ e^{\left(\frac{Cd\,\rho 0\,A\,H\,e^{(-\frac{z}{H})}}{m}\right)}$$

Now plot the results. Specify numerical values of constants.

```
> Cd:=1.7; A:=Pi; m:=2*Pi; rho0:=1.22; g:=9.81:  H:=8100.0;
  z0:=100000.0; v0:=15000.0; k:= 0.5*Cd*A*rho0*exp(-z0/(2*H));
> plot(soln2,z=0..z0,title="Speed vs.
  Altitude",thickness=3,labels=["z(m)","v(m/s)"]);
```



Note that, when atmospheric density is constant, the projectile slows down at a much greater altitude than for the variable density case. Naturally we would expect the greatest negative acceleration force to occur where the atmosphere is densest.

8.3. Problem 7.2: Numerical Solution of the Equations of Motion (no ablation).

```
> restart;
```

Define constant and initial values. For variable that change, use the list construction of Maple.

```
> rho_m:=3500:
> r:=10:
> pi:=evalf(Pi,15):
> m:=[2*pi*(r^3)*rho_m]:
> A:=pi*r*r:
> dt:=0.01:
> C_D:=1.7:
> C_L:=0.001:
> R_E:=6371000:
```

```
> rho_0:=1.22:

> H:=8100:

> g:=z->9.81*((R_E-z)/R_E)^2:

> rho_a:=z->rho_0*exp(-z/H):

> v:=[15000]:
  theta:=[45*pi/180]:
  z:=[100000]:

> tt:=[0]:
```

Start finite differencing the variables as the projectile moves through the atmosphere.

```
> for i from 1 to 1000 while z[-1] > 0 do
  z:=[op(z),z[-1]-v[-1]*sin(theta[-1]*dt)]:

> v:=[op(v),v[-1]-((C_D*rho_a(z[-1])*A*(v[-1]^2)*dt)/(2*m[-1]))+((g(z[-1
  ])*sin(theta[-1])*dt)/m[-1])]:
  theta:=
  [op(theta),
  theta[-1]+(g(z[-1])*cos(theta[-1])*dt/v[-1])-(C_L*rho_a(z[-1])*A*v[-1]
  *dt/(2*m[-1]))-(v[-1]*cos(theta[-1])*dt/(R_E+z[-1]))]:
  tt:=[op(tt),tt[-1]+dt]:
  m:=[op(m),m[-1]]
  od:
```

Plot results for various parameters.

```
> thePoints:=(j)->[tt[j],z[j]]:

> plot([seq(thePoints(j),j=1..nops(v))],title="Altitude vs.
  Time",thickness=3,labels=["Time(s)","V(m/s)"]);
```




```
> thePoints1:=(j)->[v[j],z[j]]:
> plot([seq(thePoints1(j),j=1..nops(v))],title="Altitude vs.
  Velocity",thickness=3,labels=["V(m/s)","z(m)"]);
```




```
> thePoints2:=(j)->[theta[j]*180./pi,z[j]]:
> plot([seq(thePoints2(j),j=1..nops(v))],title="Altitude vs.
  Angle",thickness=3,labels=["theta (degrees)","z(m)"]);
```




```
> thePoints3:=(j)->[m[j],z[j]]:
> plot([seq(thePoints3(j),j=1..nops(v))],title="Altitude vs.
  Mass",thickness=3,labels=["mass (kg)","z(m)"]);
```



Note that mass does not change, which is expected for this model.

8.4. Problem 7.3: Numerical Solution of the Equations of Motion with Ablation.

```
> restart;

> r:=10:
  rho_m:=3500:

> pi:=evalf(Pi,15):

> dt:=0.01:

> C_D:=1.7:

> C_L:=0.001:

> R_E:=6371000:
  C_H:=0.1:
  sigma:=5.670e-8:
  Tmax:=25000:

> rho_0:=1.22:

> H:=8100:
  Q:=8.0e6:

> A:=m->evalf(pi*(m/(2*pi*rho_m))^(2/3),15):
  g:=z->evalf(9.81*((R_E-z)/R_E)^2,15):

> rho_a:=z->evalf(rho_0*exp(-z/H),15):

> m:=[2*pi*rho_m*r^3]:
  v:=[15000]:
  theta:=[45*pi/180]:
  z:=[100000]:
  dKEdz:=0:

> for i from 1 to 1000 while z[-1] > 0 and m[-1] > 0 do
  z:=[op(z),z[-1]-v[-1]*sin(theta[-1]*dt)];
```

```
> v:=[op(v),v[-1]-((C_D*rho_a(z[-1])*A(m[-1])*(v[-1]^2)*dt)/(2*m[-1]))+(
  (g(z[-1])*sin(theta[-1])*dt)/m[-1])];
  theta:=
  [op(theta),
  theta[-1]+(g(z[-1])*cos(theta[-1])*dt/v[-1])-(C_L*rho_a(z[-1])*A(m[-1]
  )*v[-1]*dt/(2*m[-1]))-(v[-1]*cos(theta[-1])*dt/(R_E+z[-1]))];
  m:=[op(m),m[-1]-A(m[-1])*dt*min(0.5*C_H*rho_a(z[-1])*(v[-1]^3),sigma*(
  Tmax^4))/Q];
  dKEdz:=[op(dKEdz),0.5*(m[-1]*(v[-1]^2)-m[-2]*(v[-2]^2))/(z[-1]-z[-2])]
  ;
  od:
> thePoints:=(j)->[j,v[j]]:
> plot([seq(thePoints(j),j=1..nops(v))],title="Velocity vs.
Time
  steps",thickness=3,labels=["Steps","V(m/s)"]);
```




```
> thePoints1:=(j)->[v[j],z[j]]:
> plot([seq(thePoints1(j),j=1..nops(v))],title="Altitude vs.
  Velocity",thickness=3,labels=["V(m/s)","z(m)"]);
```




```
> thePoints2:=(j)->[m[j],z[j]]:
  plot([seq(thePoints2(j),j=1..nops(v))],title="Altitude vs.
  Mass",thickness=3,labels=["m(kg)","z(m)"]);
```




```
> thePoints3:=(j)->[dKEdz[j],z[j]]:
  plot([seq(thePoints3(j),j=1..nops(v))],title="Altitude vs.
  dKE/dz",thickness=3,labels=["dKE/dz(J/m)","z(m)"]);
```



8.5. **Problem 7.4: Numerical Solution of the Equations of Motion with Ablation and Deformation.**

```
> restart;
> rho_m:=3500:
  S:=1.0e7:
> pi:=evalf(Pi,15):
> dt:=0.01:
> C_D:=1.7:
> C_L:=0.001:
> R_E:=6371000:
  C_H:=0.1:
  sigma:=5.670e-8:
  Tmax:=25000:
> rho_0:=1.22:
> H:=8100:
  Q:=8.0e6:
> A:=r->evalf(pi*(r^2),15):
  g:=z->evalf(9.81*((R_E-z)/R_E)^2,15):
```

```
> rho_a:=z->evalf(rho_0*exp(-z/H),15):
  P_s:= (z,v) ->evalf(0.5*C_D*rho_a(z)*(v^2),15):
> v:=[15000]:
  r:=[10]:
  m:=[2*pi*rho_m*(r[-1]^3)]:
  theta:=[45*pi/180]:
  z:=[100000]:
  dKEdz:=0:
  fragment:=0:
> for i from 1 to 1000 while z[-1] > 0 and m[-1] > 0 do
  if(P_s(z[-1],v[-1]) <= S) and fragment=0 then
  r:=[op(r),(m[-1]/(2*pi*rho_m))^(1/3)];
  fi:
  if(P_s(z[-1],v[-1]) > S) then
  fragment:=1;
  r:=[op(r),evalf(2*r[-1]-r[-2]+(C_D*rho_a(z[-1])*(v[-1]^2)*(dt^2)/(2*rh
  o_m*r[-1])))];
  fi:
  z:=[op(z),z[-1]-v[-1]*sin(theta[-1]*dt)];
> v:=[op(v),v[-1]-((C_D*rho_a(z[-1])*A(r[-1])*(v[-1]^2)*dt)/(2*m[-1]))+(
  (g(z[-1])*sin(theta[-1])*dt)/m[-1])];
  theta:=
  [op(theta),
  theta[-1]+(g(z[-1])*cos(theta[-1])*dt/v[-1])-(C_L*rho_a(z[-1])*A(r[-1]
  )*v[-1]*dt/(2*m[-1]))-(v[-1]*cos(theta[-1])*dt/(R_E+z[-1]))];
  m:=[op(m),m[-1]-A(r[-1])*dt*min(0.5*C_H*rho_a(z[-1])*(v[-1]^3),sigma*(
  Tmax^4))/Q];
  dKEdz:=[op(dKEdz),0.5*(m[-1]*(v[-1]^2)-m[-2]*(v[-2]^2))/(z[-1]-z[-2])]
  ;
  od:
> thePoints1:=(j)->[v[j],z[j]]:
```

```
> plot([seq(thePoints1(j),j=1..nops(v))],title="Altitude vs.
  Velocity",thickness=3,labels=["V(m/s)","z(m)"]);
```




```
> thePoints2:=(j)->[m[j],z[j]]:
  plot([seq(thePoints2(j),j=1..nops(v))],title="Altitude vs.
  Mass",thickness=3,labels=["m(kg)","z(m)"]);
```



```
> thePoints3:=(j)->[dKEdz[j],z[j]]:
  plot([seq(thePoints3(j),j=1..nops(v))],title="Altitude vs.
  dKE/dz",thickness=3,labels=["dKE/dz(J/m)","z(m)"]);
```




```
> thePoints4:=(j)->[r[j],z[j]]:
  plot([seq(thePoints4(j),j=1..nops(r))],title="Altitude vs.
  Radius",thickness=3,labels=["r(m)","z(m)"]);
```



8.6. **Problem 7.5: Tunguska.** Using the information from Tables 1 and 2, the airburst altitudes (z_{ab}) are as listed below.

Table 4: Airburst Altitudes for Candidate Tunguska Objects

Object	Density (kg m^{-3})	Radius (m)	Velocity (m/s)	θ (degrees)	z_{ab} (km)
Iron	7900	22	15	45°	0
Stone	3500	29	15	45°	9
Carbonaceous	2200	34	15	45°	13
Comet	1000	32	25	45°	26

Note that the iron projectile does not airburst (although examination of the r vs. t graph indicates that it is about to immediately prior to impact).



The best agreement with the candidate Tunguska projectile appears to be for the stony asteroid, which has $z_{ab} \approx 9$ km.



9. Suggestions to the Instructor

This module is intended for use in a guided-inquiry course, where the students are introduced to the key ideas, and then encouraged to pursue the solutions to the projects in independent work.

In planning this course, you should make sure that you feel comfortable with the relevant uses of Maple (especially the symbolic solution technique in the first problem and the use of lists in the finite differencing solutions in the other problems).

This module should take a time frame of half a semester (approximately 8 weeks). At our institution, we typically teach our second course in computational science by using two modules similar to this one.

The time line for this module might resemble the following:

- Week 1: Introduction to key concepts (excluding deformation) and equations (1) to (7). Introduce students to Maple's `dsolve` utility. Students should work project 1.
- Week 2: Introduction to finite differences and numerical solutions. It would be helpful to discuss a simple Maple worksheet which uses loops and lists in the form of the projects the students will be working from now on. A useful project here would be for students to attempt to convert equations (1) to (7) into finite difference form and then into working Maple code.
- Week 3: Students work project 2.
- Week 4: Discussions, including student presentations of their work on projects 1 and 2. This would be a good point at which to engage the students in a discussion of the difference between using Maple in symbolic vs. numerical mode, and the more fundamental ideas behind pursuing a analytic vs. a numerical solution. It is important for the students to realize that a numerical solution is always inferior, but is frequently dictated when the governing differential equations cannot be analytically solved for a specific physical situation.
- Week 5: Discussion of the deformation model. Students work project 3 (as a modification of project 2).
- Week 6: Students work project 4, and then use their code to work project 5. Substantial in class discussion time might be needed to work through the logic discussed in section 7.4.
- Week 7: Continuation of work on projects 4 and 5.
- Week 8: Discussion of projects 4 and 5 including student presentations on conceptual questions.

10. Glossary of Terms

ablation: The evaporation of a hypersonic projectile caused by radiation from the shock front (see below).

atmospheric drag: The resistance force experienced by an object moving through the atmosphere. The form we use for this module is given by

$$F = -\frac{1}{2} C_D \rho_a A v^2 \tag{20}$$

The magnitude of the force is dependent on the drag coefficient C_D (which is a function of projectile shape), the atmospheric density ρ_a, the cross-sectional area of the projectile A and the square of the speed v.

airburst: An atmospheric explosion of a projectile, caused by extreme deformation (see below). During an airburst, all of the kinetic energy of the projectile is released in a very small region of the atmosphere. The effects can be large: the Tunguska airburst of 1908 devastated a forest region 2200 km^2 in area (the size of a city).

asteroid: A small object (smaller than 1000 km diameter), made of rock, iron or a mixture of the two, orbiting the Sun. Most asteroids have orbits between the planets Mars and Jupiter. Some, however, have orbits similar to the Earth's, and can collide with the Earth.

carbonaceous asteroid: A type of stony asteroid that contains substantial amounts of water and carbon compounds (such as amino acids). Carbonaceous asteroids are less strong than normal stony asteroids.

comet: A small object made predominantly of ice, orbiting the Sun. Most comets have orbits in the outer solar system (beyond the orbit of Saturn), but some have orbits that bring them into the inner solar system, where they can collide with the Earth.

deformation: When the stagnation pressure (see below) acting on a projectile exceeds its strength, the projectile will be compressed into a "pancake" shape that drastically increases its cross-sectional area. This can lead to an airburst (see above).

hypersonic: Describing a speed much greater than the speed of sound (typically ∼0.3 km/s). In this module, we consider projectiles moving at speeds of 15–25 km/s, or 45–75 times the speed of sound.

meteor: An atmospheric projectile (either a comet or asteroid) that burns up before reaching the surface.

meteorite: An atmospheric projectile (almost always an asteroid) that survives atmospheric passage to impact the surface.

scale height: The density of the Earth's atmosphere ρ_a decreases with height z, given (approximately) by the equation

$$\rho_a(z) = \rho_0 e^{-z/H} \tag{21}$$

where $\rho_0 = 1.22$ kg m^{-3}, the surface atmospheric density and $H = 8100$ m, the scale height for the atmosphere.

shock front: The name given to the dense, hot layer of air in front of a hypersonic projectile. This is formed because the air does not have sufficient time to move out of the way of the projectile. The shock front can reach temperatures hotter than the Sun's surface (6000 ◦C), and radiation from the shock front causes ablation (see above) of the projectile.

stagnation pressure: The maximum atmospheric pressure at the front surface of an atmospheric projectile,. The magnitude of the stagnation pressure is given by

$$P_s = \frac{1}{2} C_D \rho_a v^2 \tag{22}$$

If the stagnation pressure is sufficiently great, it can cause the projectile to undergo deformation and airburst. (see above).

11. References

References

[1] Allen, H. J., and A. J. Eggers, Alfred J., Jr. *A Study of the Motion and Aerodynamic Heating of Ballistic Missiles Entering the Earth's Atmosphere at High Supersonic Speeds* NACA Technical Report 1381, Forty-Fourth Annual Report of the NACA–1958. Washington, D.C.: 1959, 1125-1140. (Available at http://naca.larc.nasa.gov/reports/1958/naca-report-1381/).

[2] Baldwin, B., and Y. Shaeffer, *Ablation and breakup of large meteoroids during atmospheric entry*, J. Geophys. Res. **76**, 4653–4668 (1971).

[3] Biberman, L. M., S. Ya. Bronin and M. V. Brykin *Moving of a blunt body through a dense atmosphere under conditions of severe aerodynamic heating and ablation*, Acta. Astronaut. **7**, 53–65 (1980).

[4] Bronshten, V. A., *Physics of Meteoritic Phenomena*, Reidel, Dordrecht, 1983.

[5] Ceplecha, Z., *Earth-grazing daylight fireball of August 10, 1972*, Astron. Astrophys., **283**, 287–288 (1994).

[6] Chyba, C. F., P. J. Thomas and K. J. Zahnle, *The 1908 Tunguska explosion: atmospheric disruption of a stony asteroid*, Nature **361**, 40–44 (1993).

[7] Korycansky, D. G. and K. J. Zahnle, *High-resolution simulations of the impacts of asteroids into the venusian atmosphere III: further 3D models*, Icarus **161**, 244–261 (2003).

[8] NASA Planetary Atmospheres Node, http://atmos.nmsu.edu/atmoshome.html.

[9] Passey, Q. R. and H. J. Melosh, *Effects of atmospheric breakup on crater field formation*, Icarus **42**, 211–233 (1980).

[10] Press, W. H., S. A. Teukolsky, W. T. Vetterling and B. P. Flannery, *Numerical Recipies in C: The Art of Scientific Computing*, Cambridge University Press, (1992).

[11] Rawcliffe, R. D., C. D. Bartky, F. Li, E. Gordon, and D. Carta, *Meteor of August 10, 1972*, Nature, **2**47, 449–451 (1974).

[12] U.S. Standard Atmosphere, U.S. Government Printing Office, Washington, D.C., 1976. (Available in hard copy from the National Technical Information Office, Springfield, Virginia (Product Number: ADA-035-6000). The Fortran code can be obtained from Public Domain Aeronautical Software. A DOS executable and turbo pascal source code is available from Small World Communications).

[13] Zotkin, I. T., and M. A. Tsikulin, *Simulation of the explosion of the Tungus meteorite*, Sov. Sci. Dokl. **11**, 183–186 (1966).